Modelagem Matemática I – Prof. Eurico L. P. Ruivo

Introdução à linguagem AMPL

Objetivo: Apresentar aspectos básicos da linguagem AMPL aplicada à solução de um problema de programação linear.

Problema-exemplo (retirado de [Taha, 2008 – pp. 6 e 7])

A companhia Reddy-Mikks produz tintas para interiores e exteriores com base em duas matérias-primas, M_1 e M_2. A tabela a seguir apresenta os dados básicos do problema:

	Toneladas de matéria-prima por tonelada de:		
Matéria-prima	Tinta p/ exteriores	Tinta p/ interiores	Disponibilidade máxima diária (ton)
M_1	6	4	24
M_2	1	2	6
Lucro por tonelada (R$)	5000	4000	

Uma pesquisa de mercado indica que a demanda diária de tintas para interiores não pode ultrapassar a de tintas para exteriores por mais de 1 tonelada. Além disso, a demanda máxima diária de tinta para interiores é 2 toneladas.

A Reddy-Mikks quer determinar a mistura ótima (a melhor possível) de produtos de tintas para interiores e exteriores que maximize o lucro total diário.

Em primeiro lugar, é necessário identificarmos as **variáveis**, o **objetivo** e as **restrições** do problema.

Variáveis: Nesse caso correspondem às produções diárias (em toneladas) das tintas para exteriores e para interiores. Precisamos estabelecer um símbolo (ou nome) para cada variável. Vamos chamar de x_1 a quantidade (em toneladas) de tintas para exteriores produzida diariamente e de x_2 a quantidade (em toneladas) de tintas para interiores produzida diariamente.

Objetivo: A companhia Reddy-Mikks deseja **maximizar o lucro**. De acordo com a tabela acima, o lucro L é de R$ 5000,00 por tonelada de tinta para exteriores produzida e de R$ 4000,00 por tonelada de tinta para exteriores produzida. Assim, temos como objetivo **maximizar** $\boldsymbol{L = 5000 \cdot x_1 + 4000 \cdot x_2}$.

Restrições: Em primeiro lugar temos a disponibilidade máxima diária (em toneladas) de matéria prima por dia. Veja que devemos escrever as restrições em termos das variáveis, assim, por exemplo, o consumo da matéria-prima M_1 não pode ultrapassar 24 toneladas, sendo que cada tonelada de tinta para exteriores gasta 6 toneladas de M_1 e cada tonelada de tinta para interiores gasta 4 toneladas de M_1. Assim, usando as variáveis x_1 e x_2, podemos escrever a restrição de disponibilidade de M_1 pela inequação $6x_1 + 4x_2 \leq 24$. Da mesma maneira, a restrição de disponibilidade de M_2 é representada pela inequação $1x_1 + 2x_2 \leq 6$.

Além disso, temos uma limitação de demanda: a demanda diária de tintas para interiores não pode ultrapassar a de tintas para exteriores por mais de 1 tonelada. Isso pode ser escrito como $x_2 \leq x_1 + 1$ ou ainda como $x_2 - x_1 \leq 1$. Temos também a informação que a demanda máxima diária de tinta para interiores é de 2 toneladas. Isto é, $x_2 \leq 2$.

Outras duas restrições que não estão explícitas no enunciado é que as quantidades (em toneladas) de tintas para exteriores e para interiores produzidas não podem ser negativas. Isto é, $x_1 \geq 0$ e $x_2 \geq 0$.

Em resumo, nosso problema pode ser descrito por:

$\boldsymbol{x_1}$:quantidade (em toneladas) de tinta para exteriores produzida diariamente $\boldsymbol{x_2}$:quantidade (em toneladas) de tinta para interiores produzida diariamente
maximizar $5000\, x_1 + 4000\, x_2$
sujeito a $6x_1 + 4x_2 \leq 24$ $1x_1 + 2x_2 \leq 6$ $x_2 - x_1 \leq 1$ $x_2 \leq 2$ $x_1 \geq 0$ $x_2 \geq 0$

Uma vez que tenhamos o problema descrito em termos de suas variáveis, objetivo e restrições, podemos passar à sua modelagem via AMPL. A seguir, veremos duas maneiras de fazer isso: uma menos elaborada, que serve apenas para esse problema em questão, e outra que é construída de uma maneira que possa ser utilizada em problemas parecidos a esse.

Modelagem em AMPL 1 – Modelo específico

```
var x1>=0;
var x2>=0;
maximize L: 5000*x1+4000*x2;
subject to
c1: 6*x1+4*x2<=24;
c2: x1+2*x2<=6;
c3: x2-x1<=1;
c4: x2<=2;
solve;
display L,x1,x2;
```

As palavras destacadas em negrito correspondem a palavras-chave (termos reservados) da AMPL.
A palavra-chave **var** serve para declarar uma variável. Assim, nas duas primeiras linhas são declaradas as variáveis x_1 e x_2 juntamente com suas condições de não-negatividade (isto é, as condições $x_1 \geq 0$ e $x_2 \geq 0$).
A seguir, o objetivo é declarado: a palavra-chave **maximize** indica que é um problema de maximização. A seguir o nome da função a ser maximizada (L) é dado, seguido de dois pontos (:) indicando que ela será definida. Se o problema fosse de minimização, a palavra-chave seria **minimize**.
Em seguida, **subject to** indica que serão listadas as condições. Nas linhas de 5 a 8 constam as restrições que descrevemos anteriormente: cada uma recebe um nome único seguido de

dois pontos (:), como feito anteriormente para a função L. Note que cada linha é encerrada com um ponto-e-vírgula (;).
O comando **solve** instrui o sistema a resolver o problema descrito acima e o comando **display** faz com que os valores ótimos das variáveis indicadas após ele sejam exibidos

Copie o código acima em um editor de texto e salve o arquivo como modelo1 na pasta onde está a instalação do AMPL. Para executar o AMPL sobre esse modelo, abra o terminal do AMPL e digite **model** modelo1.txt. A saída deve ser a seguinte:

```
ampl: model modelo1.txt
MINOS 5.51: optimal solution found.
2 iterations, objective 21000
L = 21000
x1 = 3
x2 = 1.5

ampl:
```

Na saída acima temos o valor otimizado do lucro e os valores que devem ser atribuídos às variáveis x_1 e x_2 para obter o maior lucro possível. A interpretação é a seguinte: devemos produzir 3 toneladas de tintas para exteriores (x_1) e 1,5 toneladas de tintas para interiores (x_2) para obter o maior lucro possível, que será de $R\$ \ 21\ 000{,}00$ (L).

Obs.: alternativamente você pode rodar o modelo no endereço https://neos-server.org/neos/solvers/lp:bpmpd/AMPL.html, bastando fazer o upload do arquivo modelo1 no espaço indicado por AMPL Model.

Agora imagine que você precisasse preparar um modelo semelhante ao modelo acima para cada tipo de situação semelhante. Por exemplo, poderíamos ter mais tipos de tinta, mais tipos de matérias-primas e ainda outras restrições. Voltando ao quadro-resumo do problema na página anterior, é possível notar que todas as restrições e a função-objetivo tem algo em comum: são todas expressões **lineares**. Basicamente isso significa que em cada expressão só há multiplicações por constantes, adições e subtrações, sendo que nenhum outro tipo de operação é permitido.

Vamos tentar imaginar como ficaria uma versão mais geral do problema da Companhia Reddy-Mikks, construindo uma tabela semelhante à anterior, mas com p restrições $R_1, R_2, \cdots, R_p$ e q tipos distintos de tinta sendo fabricados nas quantidades $x_1, x_2, \cdots, x_q$ (em toneladas). Teríamos uma tabela como a seguinte:

	Toneladas de matéria-prima por tonelada de:				
	Tinta 1	Tinta 2	$\cdots$	Tinta q	Limite
R_1	a_{11}	a_{12}	$\cdots$	a_{1q}	d_1
R_2	a_{21}	a_{22}	$\cdots$	a_{2q}	d_2
$\vdots$	$\vdots$	$\vdots$	$\ddots$	$\vdots$	$\vdots$
R_p	a_{p1}	a_{p2}	$\cdots$	a_{pq}	d_p
Lucro	v_1	v_2	$\cdots$	v_q	

Vamos entender o conteúdo da tabela.

Os valores $d_1, d_2, \cdots, d_p$ são os limites (aqui supostos superiores) das restrições $1,2,\cdots,p$, respectivamente.

Os valores $v_1, v_2, \cdots, v_q$ correspondem a quanto cada tonelada de tinta $1,2,\cdots,q$ contribuem para o lucro L, respectivamente.

Os valores $a_{11}, a_{12}, \cdots, a_{1q}$ indicam os multiplicadores das variáveis $x_1, x_2, \cdots, x_q$ na restrição R_1, respectivamente. Os valores $a_{21}, a_{22}, \cdots, a_{2q}$ indicam o mesmo para a restrição R_2 e assim sucessivamente.

Os valores p, q, d_i, c_j e a_{ij} são chamados de **parâmetros** do modelo.

Em resumo, temos:

Parâmetro	**Variação**
p: quantidade de restrições	Livre (maior que 0)
q: quantidade de tintas	Livre (maior que 0)
d_i: limite superior da restrição i	i varia de 1 a p
v_j: contribuição da tinta j para o lucro	j varia de 1 a q
a_{ij}: multiplicador da tinta x_j na restrição R_i.	i varia de 1 a p j varia de 1 a q

Agora o quadro-resumo do problema seria:

$\boldsymbol{x_1}$: quantidade (em toneladas) de Tinta 1 produzida diariamente
$\boldsymbol{x_2}$: quantidade (em toneladas) de Tinta 2 produzida diariamente
$\vdots$
$\boldsymbol{x_q}$: quantidade (em toneladas) de Tinta q produzida diariamente

maximizar $v_1x_1 + v_2x_2 + \cdots + v_qx_q$

sujeito a

$$x_1, x_2, \cdots, x_q \geq 0$$
$$c_1: a_{11}x_1 + a_{12}x_2 + \cdots + a_{1q}x_q \leq d_1$$
$$c_2: a_{21}x_1 + a_{22}x_2 + \cdots + a_{2q}x_q \leq d_2$$
$$\vdots$$
$$c_p: a_{p1}x_1 + a_{p2}x_2 + \cdots + a_{pq}x_q \leq d_q$$

As somas na tabela acima poderiam ser reescritas utilizando-se o símbolo de somatório (Σ):

$\boldsymbol{x_1}$: quantidade (em toneladas) de Tinta 1 produzida diariamente
$\boldsymbol{x_2}$: quantidade (em toneladas) de Tinta 2 produzida diariamente
$\vdots$
$\boldsymbol{x_q}$: quantidade (em toneladas) de Tinta q produzida diariamente

maximizar $\sum_{j=1}^{q} v_jx_j$

sujeito a

$$x_1, x_2, \cdots, x_q \geq 0$$
$$c_1: \sum_{j=1}^{q} a_{1j}x_j \leq d_1$$

$$c_2: \sum_{j=1}^{q} a_{2j}x_j \leq d_2$$

$$\vdots$$

$$c_p: \sum_{j=1}^{q} a_{pj}x_j \leq d_p$$

Podemos escrever esse modelo mais geral em AMPL da seguinte maneira:

Modelagem em AMPL 2 – Modelo algébrico

```
param p;
param q;
param v{1..q};
param d{1..p};
param a{1..p,1..q};

var x{1..q};

maximize L: sum{j in 1..q}v[j]*x[j];

subject to r{i in 1..p}: sum{j in 1..q}a[i,j]*x[j]<=d[i];
```

O termo **param** indica que um objeto é um *parâmetro* do problema, isto é, ele não é uma variável do problema, mas pode variar de problema para problema. Os parâmetros p e q são simplesmente números, então são declarados apenas precedidos do termo **param**, como consta nas linhas 1 e 2.

Já no caso dos $v_1, v_2, \cdots, v_q$, são uma lista de parâmetros que depende ela própria do valor do parâmetro q. Para declará-la como uma lista com elementos de 1 a q utiliza-se a construção **param** v{1..q}. O mesmo é feito para a lista de parâmetros $d_1, d_2, \cdots, d_p$, mas agora com os elementos indo de 1 a p: **param** d{1..p}.

Agora, para os multiplicadores a_{ij}, temos dois índices: o primeiro (i) varia de 1 a p (que é a quantidade de restrições) e o segundo (j) varia de 1 a q (que é a quantidade de variáveis, no caso, de tintas). Assim o parâmetro a será uma matriz, com p linhas e q colunas, o que é declarado por **param** a{1..p,1..q}.

Na sequência, o objetivo começa a ser declarado com o termo **maximize** e o nome da função L seguido de dois pontos e sua definição: a soma dos termos $v_1x_1, v_2x_2, \cdots, v_qx_q$. Isso é feito por meio da palavra-chave **sum** (que indica uma soma) e da indicação do índice que varia na soma escrito como {j **in** 1..q}, indicando que o índice da soma é j e que varia de 1 a q. Por fim escreve-se o que será somado (no caso o produto v_jx_j) indicado por $v[j] * x[j]$. Assim, a função $L = \sum_{j=1}^{q} v_jx_j$ é escrita em AMPL como L:sum{j in 1..q}v[j]*x[j].

Por fim, as restrições $R_1, R_2, \cdots, R_p$ são enunciadas de acordo com as notações definidas anteriormente: a lista de restrições r é declarada logo após o **subject to** como r{i **in** 1..p} e cada uma delas é definida como **sum**{j in 1..q}a[i,j]*x[j]<=d[i].

Salve esse arquivo na pasta do AMPL como modelo2.

Observe que agora não escrevemos o comando **solve** nem o **display**. Isso porque acabamos de criar um arquivo de modelo e não de um problema particular. Isto é, criamos um modelo geral, mas que só poderá ser resolvido de fato se fornecermos os valores dos *parâmetros*.

Vamos agora escrever o problema da Companhia Reddy-Mikks num arquivo separado. Lembre-se que precisamos fornecer os valores de todos os parâmetros de nosso modelo.

Para facilitar, vamos reescrever o objetivo e as restrições do problema da Companhia Reddy-Mikks aqui, desconsiderando a não-negatividade das variáveis que já foi implementada acima no modelo:

Maximizar $L = 5000\,x_1 + 4000\,x_2$

Sujeito a:

$$r_1: 6x_1 + 4x_2 \leq 24$$

$$r_2: 1x_1 + 2x_2 \leq 6$$

$$r_3: -1x_1 + 1x_2 \leq 1$$

$$r_4: 0x_1 + 1x_2 \leq 2$$

Modelagem em AMPL 2 – Dados do problema
data; **param** p:=4; **param** q:=2; **param** v:= **1** 5000 **2** 4000; **param** d:= **1** 24 **2** 6 **3** 1 **4** 2; **param** a: **1 2** := **1** 6 4 **2** 1 2 **3** -1 1 **4** 0 1; **solve**; **display** L, x;
A palavra-chave **data** indica que o que está escrito a seguir são os dados de um problema. Para definir os valores de um parâmetro utilizamos dois-pontos seguido de um igual. Assim a linha **param** p:=4; indica que estamos atribuindo o valor 4 para o parâmetro p, que indica a quantidade de restrições. Da mesma maneira **param** q:= 2; indica que o parâmetro q que recebe o número de variáveis será igual a 2 (no problema da Reddy-Mikks isso significa duas tintas diferentes). Para atribuição de valores a um vetor/lista, a notação é um pouco diferente. Os números destacados em negrito no código acima correspondem à posição (índice) do elemento na lista. Assim, **param** v:= **1** 5 **2** 4; indica que o primeiro elemento da lista v recebe valor 5000 e que o segundo elemento da lista v recebe valor 4000. Isto é, ele declara $v = [5000\ 4000]$. O mesmo ocorre para a lista d, dada por $d = [24\ 6\ 1\ 2]$, que corresponde aos valores à direita das desigualdades dadas pelas restrições. Por fim, a matriz a com os multiplicadores das variáveis em cada desigualdade é fornecida. A estrutura a: **1 2** := indica que cada elemento depois do igual tem duas coordenadas separadas por um espaço. Dito de outra maneira, a: **1 2** := indica que a é uma matriz com duas colunas. Agora sua primeira linha é especificada por **1** 6 4 sua segunda linha por **2** 1 2 e

assim sucessivamente, tomando as constantes multiplicando x_1 seguidas das constantes multiplicando x_2 em cada restrição. Isso fornece a matriz $(a_{ij}) = \begin{bmatrix} 6 & 4 \\ 1 & 2 \\ -1 & 1 \\ 0 & 1 \end{bmatrix}$.

Por fim, os comandos **solve** e **display** são usados conforme discutido anteriormente.

Salve o código contendo os dados do problema no arquivo modelo2dados.

Abra o terminal do AMPL e digite:

model modelo2.txt;

data modelo2dados.txt;

(Caso os arquivos não estejam na pasta do AMPL, escreva o caminho completo de cada arquivo).

Os comandos acima devem gerar a seguinte saída, que é a mesma obtida pelo modelo específico:

```
ampl: model modelo2.txt
ampl: data modelo2dados.txt
MINOS 5.51: optimal solution found.
2 iterations, objective 21000
L = 21000

x [*] :=
1  3
2  1.5
;
```

Alternativamente, você pode rodar o modelo no endereço https://neos-server.org/neos/solvers/lp:bpmpd/AMPL.html, bastando fazer o upload do arquivo modelo2 no espaço indicado por AMPL Model e do arquivo modelo2dados no espaço indicado por AMPL Data.

Problema Proposto (adaptado de [Taha,2008]): Suponha que a Companhia Reddy-Mikks passou agora a produzir um terceiro tipo de tinta, denominado “tinta para veículos”. Para fabricar uma tonelada de Tinta para Veículos, utiliza-se 0,5 toneladas da matéria-prima M_1 e 0,75 toneladas da matéria-prima M_2. A demanda diária dessa tinta está entre 0,5 toneladas e 1,5 toneladas e o lucro por tonelada é de R$ 3 500,00.

Mantendo as demais condições do problema original, modifique os arquivos modelo1.txt e modelo2dados.txt para que eles incluam essa nova tinta no modelo. Salve-os com o nome [TIA]_modelo1.txt e [TIA]_modelo2dados.txt, onde [TIA] deve ser substituído por seu TIA. O arquivo [TIA]_modelo1.txt deve conter uma modelagem específica para o problema e o arquivo [TIA]_modelo2dados.txt deve conter os dados do problema de maneira que ele possa ser resolvido pelo modelo implementado no arquivo modelo2.txt.

A entrega dos dois arquivos deve ser feita via Moodle de acordo com as especificações acima e também com aquelas descritas na entrega no Moodle.